# Maintainer: Bruno Pagani <archange@archlinux.org>
# Maintainer: Caleb Maclennan <caleb@alerque.com>

pkgname=pqiv
pkgver=2.12
pkgrel=6
pkgdesc='Powerful image viewer with minimal UI'
arch=(loong64 x86_64)
url="https://github.com/phillipberndt/pqiv"
license=(GPL3)
depends=(gtk3 gdk-pixbuf2 glib2 cairo)
optdepends=(
    'ffmpeg: rudimentary video support'
    'libarchive: archives and cbX comic book support'
    'libspectre: PS/EPS support'
    'libwebp: WebP support'
    'imagemagick: support for various images formats like PSD'
    'poppler-glib: rudimentary PDF support'
)
makedepends=(ffmpeg libarchive libspectre libwebp imagemagick poppler-glib)
source=(${url}/archive/${pkgver}/${pkgname}-${pkgver}.tar.gz)
sha256sums=('1538128c88a70bbad2b83fbde327d83e4df9512a2fb560eaf5eaf1d8df99dbe5')

build() {
  cd ${pkgname}-${pkgver}
  ./configure --backends-build=shared
  make
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" install
}
